

```
% W-волк, G-коза, C-капуста, F-фермер
% Состояние: 1-левый берег, 0-правый берег
% конечное состояние - все на правом берегу
route([0,0,0,0],P) :- writeln(P),!.
route(Sost_x,P) :-
  next(Sost_x,Sost_y), % выбор приемника
  not(member(Sost_y,P)), % исключение повторного посещения
  route(Sost_y,[Sost_x|P]). % переход к следующему состоянию
% выбор приемника
next(Sost_x,Sost_y) :- move(Sost_x,Sost_y),
  not(conflict(Sost_y)).
% возможные перемещения
move([1,G,C,1], [0,G,C,0]). % везем вправо волка
move([W,1,C,1], [W,0,C,0]). % везем вправо козу
move([W,G,1,1], [W,G,0,0]). % везем вправо капусту
move([W,G,C,1], [W,G,C,0]). % порожняком идем вправо
move([W,G,C,0], [W,G,C,1]). % порожняком идем влево
move([0,G,C,0], [1,G,C,1]). % везем влево волка
move([W,0,C,0], [W,1,C,1]). % везем влево козу
move([W,G,0,0], [W,G,1,1]). % везем влево капусту
% конфликтные состояния
conflict([1,1,_,0]). % фермер-справа, слева - волк, коза
conflict([_,1,1,0]). % фермер-справа, слева - коза, капуста
conflict([0,0,_,1]). % фермер-слева, справа - волк, коза
conflict([_,0,0,1]). % фермер-слева, справа - коза, капуста

%Поиск решения
%:- route([1,1,1,1],[]).
```

juna@artamonov: ~

Файл Правка Вид Поиск Терминал Справка

```
juna@artamonov:~$ swipl t.pl
Welcome to SWI-Prolog (threaded, 64 bits, version 7.6.4)
SWI-Prolog comes with ABSOLUTELY NO WARRANTY. This is free software.
Please run ?- license. for legal details.

For online help and background, visit http://www.swi-prolog.org
For built-in help, use ?- help(Topic). or ?- apropos(Word).

?- route([1,1,1,1],[]).
[[0,1,0,1],[0,1,0,0],[0,1,1,1],[0,0,1,0],[1,0,1,1],[1,0,1,0],[1,1,1,1]]
true .

?- 
```